NPS-MA-91-013

NAVAL POSTGRADUATE SCHOOL

Monterey, California

QUALITY CONTROL OF METEOROLOGICAL OBSERVATIONS

by

Toke Jayachandran
Richard Franke

Technical Report for Period

January 1991-June 1991

Approved for public release; distribution unlimited

Prepared for: Naval Oceanographic and Atmospheric Research Laboratory, Atmospheric Directorate
Monterey, CA 93943

NAVAL POSTGRADUATE SCHOOL
MONTEREY CALIFORNIA 93943

Rear Admiral R. W. West, Jr.
Superintendent

Harrison Shull
Provost

This report was prepared in conjunction with research conducted for the Naval Postgraduate School and funded by the Naval Oceanographic and Atmospheric Research Laboratory, Atmospheric Directorate.

Prepared by:

TOKE JAYACHANDRAN
Professor of Mathematics

Richard Franke

RICHARD FRANKE
Professor of Mathematics

Reviewed by:

HAROLD M. FREDRICKSEN
Chairman
Department of Mathematics

P. J. Marto

PAUL J. MARTO
Dean of Research

UNCLASSIFIED
SECURITY CLASSIFICATION OF THIS PAGE

REPORT DOCUMENTATION PAGE

Form Approved OMB No 0704-0188

1a REPORT SECURITY CLASSIFICATION UNCLASSIFIED	1b RESTRICTIVE MARKINGS
2a SECURITY CLASSIFICATION AUTHORITY	3 DISTRIBUTION / AVAILABILITY OF REPORT Approved for public release; distribution unlimited
2b DECLASSIFICATION / DOWNGRADING SCHEDULE	
4 PERFORMING ORGANIZATION REPORT NUMBER(S) NPS-53-91-013	5 MONITORING ORGANIZATION REPORT NUMBER(S)
6a NAME OF PERFORMING ORGANIZATION Naval Postgraduate School 6b OFFICE SYMBOL (If applicable) MA	7a NAME OF MONITORING ORGANIZATION Naval Oceanographic and Atmospheric Research Lab., Atmos. Dir.
6c ADDRESS (City, State, and ZIP Code) Monterey, CA 93943	7b ADDRESS (City, State, and ZIP Code) Monterey, CA 93943
8a NAME OF FUNDING / SPONSORING ORGANIZATION Naval Ocean. & Atmos. Research Lab., Atmos. Dir. 8b OFFICE SYMBOL (If applicable)	9 PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER 0602435N
8c ADDRESS (City, State, and ZIP Code) Monterey, CA 93943	10 SOURCE OF FUNDING NUMBERS PROGRAM ELEMENT NO / PROJECT NO / TASK NO / WORK UNIT ACCESSION NO

11 TITLE (Include Security Classification)

Quality Control of Meteorological Observations (U)

12 PERSONAL AUTHOR(S)

Toke Jayachandran and Richard Franke

13a TYPE OF REPORT	13b TIME COVERED	14 DATE OF REPORT (Year, Month, Day)	15 PAGE COUNT
Technical Report	FROM 01/91 TO 06/91	16 September 1991	19

16 SUPPLEMENTARY NOTATION

17 COSATI CODES			18 SUBJECT TERMS (Continue on reverse if necessary and identify by block number)
FIELD	GROUP	SUB-GROUP	radiosonde data, meteorological data, outliers observation increments, error bias, quality control

19 ABSTRACT (Continue on reverse if necessary and identify by block number)

A test for outliers is discussed and a computer program for its application to meteorological data for purposes of quality control is given. The data for the examples is from the NOARL database of observation minus forecast increments. The program will tag outliers from a group of stations based on daily values over a period of time, mean values over that time period, and standard deviations over that time period.

20 DISTRIBUTION / AVAILABILITY OF ABSTRACT ☒ UNCLASSIFIED/UNLIMITED ☐ SAME AS RPT ☐ DTIC USERS	21 ABSTRACT SECURITY CLASSIFICATION UNCLASSIFIED	
22a NAME OF RESPONSIBLE INDIVIDUAL Toke Jayachandran	22b TELEPHONE (Include Area Code) 408-646-2600	22c OFFICE SYMBOL MA/JY

DD Form 1473, JUN 86 Previous editions are obsolete SECURITY CLASSIFICATION OF THIS PAGE UNCLASSIFIED

S/N 0102-LF-014-6603

QUALITY CONTROL OF METEOROLOGICAL OBSERVATIONS

Toke Jayachandran and Richard Franke

Introduction: A major input component to weather prediction systems is the observational data, typically geopotential heights and winds from weather observing stations around the world. The Navy Operational Global Atmospheric Prediction System (NOGAPS) produces weather forecasts using a spectral forecast model. Preceding the forecast is a multivariate objective analysis, followed by a nonlinear normal mode initialization scheme. The multivariate objective analysis scheme uses the weather observations to produce analyzed fields with minimum statistical error. As is to be expected, some of the observations will turn out to be erroneous for various reasons such as the mis-calibration of the measurement devices, failure to perform prescribed data corrections, communications errors and simple recording errors. We present in this paper a quality control methodology to screen the data to identify those observations that appear to be erroneous in a statistical sense. It is then up to a meteorological analyst or a decision making algorithm to examine the tagged observations and decide on a course of action - delete the miscreant observations, modify the data if appropriate or include the observations, as is, for NOGAPS analysis. The procedures can also be used to monitor the reliability of the observing stations and to alert them of any mispractices.

Data Quality Tests: The methodology proposed here is an adaptation of a statistical test for outliers (Dixon, 1950 [1]). The first step in the application of this test is to partition the monitoring stations into small "contiguous groups". The recorded observations within any single group are expected to be "similar" because the weather patterns at these monitoring sites are comparable due to the proximity of the stations to each other.

One approach towards identifying such a group of stations is the following. Select the latitude and longitude for a target location on the surface of the earth (this could be the coordinates of a reporting station of interest) and choose an approximately square region by specifying a range for the latitude, say 5^0 above and below the target, and within a similar distance in the longitudinal direction. The observational increments (differences between the reported measurements and the forecast values) for all of the stations within the designated area are then subjected to a sequence of outlier tests.

Assume that the number of reporting stations within the target area is N and the data from the most recent k days (say 30 days) is to be examined for quality. Let X_{i1} , X_{i2} , ... , X_{ik} be the observation increments for the i^{th} station, i = 1, 2, . . . , N; let M_1 , M_2 , ... , M_N be the means of the increments (averaged over the k days for each of the N stations) and S_1, S_2 , . . . , S_N the standard deviations (over the k days) for the N stations.

The first test we propose is to compare the daily observational increments X_{1j} , X_{2j} , ... , X_{Nj} for the j^{th} day, and check for "outliers". Let $Y_1 < Y_2 < . . . < Y_N$ be the data rearranged in increasing order. Compute

$$r_{11} = (Y_N - Y_{N-1}) \; / \; (Y_N - Y_2) \; ;$$

if r_{11} exceeds the tabulated value in Table A (extracted from Dixon,1950), corresponding to a chosen statistical significance level ß (say .05), conclude that the observed datum for the station corresponding to Y_N is an outlier. This implies that the

observation increment Y_N is too large relative to the increments for the other stations in the region. That is, the observed geopotential height appears to be too high (large) when compared to the forecast. Now compute

$$r'_{11} = (Y_2 - Y_1) / (Y_{N-1} - Y_1)$$

and conclude that Y_1 is an outlier if r'_{11} exceeds the tabulated value in Table A; if this happens,the implication is that the observed geopotential height is too low (small) compared to the forecast. These tests can be performed for each of the days for which data is available. The meteorological analyst or the decision making algorithm will then have to choose a course of action, based on some established decision rule, for each of the observations failing the test. The two tests were applied to actual data from three different areas of the globe viz., US, Europe, and China. In each area we identified a location by specifying its latitude and longitude and then selected all the stations within 4^0 of this location. The results of the daily tests are presented in Tables 1a,b,c. The I.D. numbers of the stations included for the test are listed in the first column. The entries (there are two entries in each of the 16 columns) in the table represent the number of days in October '90 on which the stations failed the outlier test at the low end (r_{11}' test) and the test at the high end (r_{11} test) using the data (geopotential height differences) for 16 pressure levels (1000, 850, 700, 500, 400, 300, 250, 200, 150, 100, 70, 50, 30, 20, 15, 10 mb). A dot in any position implies no failures; an asterisk indicates insufficient data to perform the tests. Such a table can be used in making quality judgments on the performance of the monitoring stations.

Another test that can also be used to monitor the performance

quality of weather stations is to perform the two outlier tests on the averages M_1 , M_2 , ... , M_N . The data for the tests are these averages rearranged in increasing order. Tables 2a,b,c present the same type of information as Tables 1a,b,c except that this time the entries are a 1 or a dot; a 1 signifies that the average for a station failed the outlier test while a dot would signify that the station passed the test.

A third outlier test, that can be viewed as a test for consistency, is to check for an outlier in the station standard deviations S_1 , S_2 , ... , S_N . If $Y_1 < Y_2 < . . . < Y_N$ are these standard deviations in ordered form, compute

$$r_{10} = (Y_N - Y_{N-1}) \ / \ (Y_N - Y_1)$$

and conclude that the standard deviation corresponding to Y_N is statistically too large relative to the others, if r_{10} exceeds the critical value in Table B (extracted from Dixon, 1950) at a significance level ß (say .05). Tables 3a,b,c are similar to Tables 1 and 2 using the station standard deviations, with the difference that only the high end outlier test is used.

We have written a computer program, using Fortran, that will select (1) the reporting stations within a specified (approximately square) region around a specified location and (2) the appropriate data for these stations from the Naval Oceanographic and Atmospheric Research Laboratory (NOARL) files [2], and automatically performs all of the tests described above. Tables 1, 2 and 3, described above, are the actual outputs of this program. The program also generates additional information such as the number of times the daily outlier tests were performed (Table 4) and the number of days for which data was available (Table 5), during the period specified for the test. The numbers

in Table 4 will sometimes be smaller than the corresponding entries in Table 5; this happens whenever the number of data points available for the test is smaller than the minimum sample size of 4. Table 6 is an example of another type of output of the program; for a selected station the table provides the daily test history, over the 16 pressure levels. A "PASS" indicates that the station passed all the daily tests; a "LFAIL" signifies that the observational increment is too low as compared to the numbers for the other stations in the region, on at least one day of the testing period. A "FAILH" denotes that the increment for at least one day was deemed to be high. Similar tables displaying the results of the means and standard deviations tests are also produced. In the "TOTAL" column a "PASS" means that the station passed all the tests on all the days and a "FAIL" implies the converse.

References

[1] Dixon, W.J., "Ratios Involving Extreme Values", Annals of Mathematical Statistics, Vol.22, No.1, March 1951.

[2] Baker, Nancy L., "The Adaptive Quality Control and Analysis Statistical Database", NOARL Technical Note, May 29, 1991, Naval Oceanographic and Atmospheric Research Laboratory, Stennis Space Center, Bay St Louis, Miss 39529.

TABLE A

$Pr(r_{10} > R) = \alpha$

n \ α	.005	.01	.02	.05	.10	.20	.30	.40	.50	.60	.70	.80	.90	.95	α / n
3	.994	.988	.976	.941	.886	.781	.684	.591	.500	.409	.316	.219	.114	.059	3
4	.926	.889	.846	.765	.679	.560	.471	.394	.324	.257	.193	.130	.065	.033	4
5	.821	.780	.729	.642	.557	.451	.373	.308	.250	.196	.146	.097	.048	.023	5
6	.740	.698	.644	.560	.482	.386	.318	.261	.210	.164	.121	.079	.038	.018	6
7	.680	.637	.586	.507	.434	.344	.281	.230	.184	.143	.105	.068	.032	.016	7
8	.634	.590	.543	.468	.399	.314	.255	.208	.166	.128	.094	.060	.029	.014	8
9	.598	.555	.510	.437	.370	.290	.234	.191	.152	.118	.086	.055	.026	.013	9
10	.568	.527	.483	.412	.349	.273	.219	.178	.142	.110	.080	.051	.025	.012	10
11	.542	.502	.460	.392	.332	.259	.208	.168	.133	.103	.074	.048	.023	.011	11
12	.522	.482	.441	.376	.318	.247	.197	.160	.126	.097	.070	.045	.022	.011	12
13	.503	.465	.425	.361	.305	.237	.188	.153	.120	.092	.067	.043	.021	.010	13
14	.488	.450	.411	.349	.294	.228	.181	.147	.115	.088	.064	.041	.020	.010	14
15	.475	.438	.399	.338	.285	.220	.175	.141	.111	.085	.062	.040	.019	.010	15
16	.463	.426	.388	.329	.277	.213	.169	.136	.107	.082	.060	.039	.019	.009	16
17	.452	.416	.379	.320	.269	.207	.165	.132	.104	.080	.058	.038	.018	.009	17
18	.442	.407	.370	.313	.263	.202	.160	.128	.101	.078	.056	.036	.018	.009	18
19	.433	.398	.363	.306	.258	.197	.157	.125	.098	.076	.055	.036	.017	.008	19
20	.425	.391	.356	.300	.252	.193	.153	.122	.096	.074	.053	.035	.017	.008	20
21	.418	.384	.350	.295	.247	.189	.150	.119	.094	.072	.052	.034	.016	.008	21
22	.411	.378	.344	.290	.242	.185	.147	.117	.092	.071	.051	.033	.016	.008	22
23	.404	.372	.338	.285	.238	.182	.144	.115	.090	.069	.050	.033	.016	.008	23
24	.399	.367	.333	.281	.234	.179	.142	.113	.089	.068	.049	.032	.016	.008	24
25	.393	.362	.329	.277	.230	.176	.139	.111	.088	.067	.048	.032	.015	.008	25
26	.388	.357	.324	.273	.227	.173	.137	.109	.086	.066	.047	.031	.015	.007	26
27	.384	.353	.320	.269	.224	.171	.135	.108	.085	.065	.047	.031	.015	.007	27
28	.380	.349	.316	.266	.220	.168	.133	.106	.084	.064	.046	.030	.015	.007	28
29	.376	.345	.312	.263	.218	.166	.131	.105	.083	.063	.046	.030	.014	.007	29
30	.372	.341	.309	.260	.215	.164	.130	.103	.082	.062	.045	.029	.014	.007	30

TABLE B

$Pr(r_{11} > R) = \alpha$

n \ α	.005	.01	.02	.05	.10	.20	.30	.40	.50	.60	.70	.80	.90	.95	α / n
4	.995	.991	.981	.955	.910	.822	.737	.648	.554	.459	.362	.250	.131	.069	4
5	.937	.916	.876	.807	.728	.615	.524	.444	.369	.296	.224	.151	.078	.039	5
6	.839	.805	.763	.689	.609	.502	.420	.350	.288	.227	.169	.113	.056	.028	6
7	.782	.740	.689	.610	.530	.432	.359	.298	.241	.189	.140	.093	.045	.022	7
8	.725	.683	.631	.554	.479	.385	.318	.260	.210	.164	.121	.079	.037	.019	8
9	.677	.635	.587	.512	.441	.352	.288	.236	.189	.148	.107	.070	.033	.016	9
10	.639	.597	.551	.477	.409	.325	.265	.216	.173	.134	.098	.063	.030	.014	10
11	.606	.566	.521	.450	.385	.305	.248	.202	.161	.124	.090	.058	.028	.013	11
12	.580	.541	.498	.428	.367	.289	.234	.190	.150	.116	.084	.055	.026	.012	12
13	.558	.520	.477	.410	.350	.275	.222	.180	.142	.109	.079	.052	.025	.012	13
14	.539	.502	.460	.395	.336	.264	.212	.171	.135	.104	.075	.049	.024	.011	14
15	.522	.486	.445	.381	.323	.253	.203	.164	.129	.099	.072	.047	.023	.011	15
16	.508	.472	.432	.369	.313	.244	.196	.158	.124	.095	.069	.045	.022	.011	16
17	.495	.460	.420	.359	.303	.236	.190	.152	.119	.092	.067	.044	.021	.010	17
18	.484	.449	.410	.349	.295	.229	.184	.148	.116	.089	.065	.042	.020	.010	18
19	.473	.439	.400	.341	.288	.223	.179	.143	.112	.087	.063	.041	.020	.010	19
20	.464	.430	.392	.334	.282	.218	.174	.139	.110	.084	.061	.040	.019	.010	20
21	.455	.421	.384	.327	.276	.213	.170	.136	.107	.082	.059	.039	.019	.009	21
22	.446	.414	.377	.320	.270	.208	.166	.132	.104	.081	.058	.038	.018	.009	22
23	.439	.407	.371	.314	.265	.204	.163	.130	.102	.079	.056	.037	.018	.009	23
24	.432	.400	.365	.309	.260	.200	.160	.127	.100	.077	.055	.036	.018	.009	24
25	.426	.394	.359	.304	.255	.197	.156	.124	.098	.076	.054	.036	.017	.009	25
26	.420	.389	.354	.299	.250	.193	.154	.122	.096	.074	.053	.035	.017	.008	26
27	.414	.383	.349	.295	.246	.190	.151	.120	.095	.073	.052	.034	.017	.008	27
28	.409	.378	.344	.291	.243	.188	.149	.118	.093	.072	.051	.034	.016	.008	28
29	.404	.374	.340	.287	.239	.185	.146	.116	.092	.070	.051	.033	.016	.008	29
30	.399	.369	.336	.283	.236	.182	.144	.115	.090	.069	.050	.032	.016	.008	30

Extracted from: "Ratios of Extrema Values", W. J. Dixon, Annals of Mathematical Statistics. Vol. 22, No. 1, March, 1951.

STATION / Lvl	1		2		3		4		5		6		7		8		9		10		11		12		13		14		15		16	
72456	.	.	.	.	.	.	.	2	.	.	1	.	1	.	1	1	3	.	.	.	2	.	.	.	2	.	.	1	1	2	*	*
72357	.	.	2	.	5	.	2	.	1	.	1	.	2	.	1	.	1	.	1	.	2	.	.	.	2	.	1	.	1	.	*	*
72349	.	1	1	.	.	.	.	.	.	.	.	.	.	1	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*
72340	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	1	1	.	.	.	.	.	.	.	.	.	.	.	1	.	*	*
72260	.	2	.	.	.	.	.	.	.	.	.	1	.	1	2	1	.	.	.	.	.	.	.	1	.	1	.	.	.	2	*	*
72247	1	.	1	1	.	1	.	.	.	.	.	.	1	.	.	.	1	.	.	.	1	.	1	.	1	.	3	.	5	.	*	*
72239	.	.	.	.	.	1	.	.	.	.	.	.	.	.	.	.	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
72257	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*

Table 1a: Failures in daily outlier test (. = 0, * = no test)
U.S. stations near (35^{o},-95^{o})
10/01/90 to 10/31/90 at 00Z

STATION / Lvl	1		2		3		4		5		6		7		8		9		10		11		12		13		14		15		16	
26702	.	1	.	1	.	1	.	3	.	1	.	.	.	.	.	.	.	.	.	1	.	1	.	1	.	1	.	2	.	1	.	.
26406	1	.	1	.	2	.	1	.	1	.	1	.	.	3	.	.	.	1	.	.	.	.	.	.	.	.	.	.	.	.	1	.
12425	.	.	1	.	.	1	.	.	.	.	.	.	1	.	.	1	.	.	.	.	.	1	.	2	1	2	1	1	.	.	.	.
12374	.	.	.	.	.	.	.	1	.	1	.	.	.	.	.	1	.	1	.	1	.	1	.	.	.	2	.	2	.	1	.	.
12330	.	.	1	1	.	.	.	.	.	1	.	.	.	.	.	.	.	.	.	1	.	.	.	.	.	1	.	1	*	*	*	*
12120	.	.	.	.	.	.	.	.	1	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*	*	*
10437	.	1	.	1	.	.	.	.	.	1	.	.	.	.	.	.	.	.	.	.	.	.	*	*	*	*	*	*	*	*	*	*
10338	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
10046	1	.	1	.	2	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*	*	*	*	*	*	*	*	*
10035	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	1	.	.	.
9393	.	.	.	.	.	.	.	.	1	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	1	.	.	.
9184	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	1	.	.	.	1	.	.	.	.	.	.	.	.	.	.	.
6181	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
2591	.	.	.	.	.	.	.	.	1	.	2	.	1	.	1	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*	*	*
2527	.	.	.	.	.	.	.	.	1	.	1	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*	*	*

Table 1b: Failures in daily outlier test (. = 0, * = no test)
European stations near (55^{o},15^{o})
10/01/90 to 10/31/90 at 00Z

STATION / Lvl	1		2		3		4		5		6		7		8		9		10		11		12		13		14		15		16	
57679	1	.	.	.	.	.	.	1	.	1	.	2	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*
57516	*	*	.	1	.	1	.	.	1	.	.	.	.	.	.	.	.	1	.	.	1	2	.	1	.	1	.	1	.	1	*	*
57494	.	1	1	.	2	.	1	1	.	.	1	.	1	.	1	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*
57461	.	.	*	*	.	.	1	.	.	.	.	.	.	.	.	.	.	.	.	.	1	.	.	.	.	.	.	.	.	.	*	*
57447	.	3	*	*	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*
57328	*	*	*	*	.	.	1	.	1	.	.	.	.	.	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
57178	.	.	*	*	1	.	.	2	.	1	1	.	.	1	1	2	.	3	.	1	.	1	.	.	.	.	.	.	.	.	*	*
57127	.	.	2	1	.	.	1	1	.	1	.	.	.	.	.	.	.	.	.	.	.	.	.	1	.	.	.	.	.	1	*	*
57083	.	.	.	1	.	.	.	1	.	.	.	.	.	1	.	.	.	.	.	.	.	.	.	.	.	.	1	.	.	.	*	*
57036	.	4	1	.	1	1	1	.	.	1	1	.	.	.	.	.	1	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*
57245	*	*	*	*	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*	*	*	*	*	*	*	*	*
57633	*	*	*	*	*	*	1	.	.	.	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*

Table 1c: Failures in daily outlier test (. = 0, * = no test)
Chinese stations near (32^{o},110^{o})
10/01/90 to 10/31/90 at 12Z

```
STATION / Lvl 1   2   3   4   5   6   7   8   9   10  11  12  13  14  15  16

 72456        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 72357        . . . . 1 . 1 . . . . . . . . . . . . . . . . . . . . . . . . .
 72349        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 72340        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 72260        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 72247        . . . . . . . . . . . . . . . . . . . . . . . . . . 1 . 1 . . .
 72239        * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
 72257        * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
```

Table 2a: Failures in the means test (. = 0, * = no test)
U.S. stations near (35^{o},-95^{o})
10/01/90 to 10/31/90 at 00Z

```
STATION / Lvl 1   2   3   4   5   6   7   8   9   10  11  12  13  14  15  16

 26702        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 26406        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 12425        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 12374        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . * *
 12330        . . . . . . . . . . . . . . . . . . . . . . . . . . . . * * * *
 12120        . . . . . . . . . . . . . . . . . . . . . . . . . . * * * * * *
 10437        . . . . . . . . . . . . . . . . . . . . . . * * * * * * * * * *
 10338        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 10046        . . . . . . . . . . . . . . . . . . . . . . * * * * * * * * * *
 10035        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
  9393        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
  9184        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
  6181        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
  2591        . . . . . . . . . . . . . . . . . . . . . . . . . . . . * * * *
  2527        . . . . . . . . . . . . . . . . . . . . . . . . . . . . * * * *
```

Table 2b: Failures in the means test (. = 0, * = no test)
European stations near (55^{o},15^{o})
10/01/90 to 10/31/90 at 00Z

```
STATION / Lvl 1   2   3   4   5   6   7   8   9   10  11  12  13  14  15  16

 57679        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . * *
 57516        * * . 1 . . . . . . . . . . . . . 1 . . . . . . . 1 . 1 . 1 . .
 57494        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 57461        . . * * . . . . . . . . . . . . . . . . . . . . . . . . . . * *
 57447        . . * * . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 57328        * * * * . . . . . . 1 . . . * * * * * * * * * * * * * * * * * *
 57178        . . * * . . . . . . . . . . . . . . . . . . . . . . . . . . * *
 57127        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
 57083        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . * *
 57036        . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . * *
 57245        * * * * . . . . . . . . . . . . . . . . . . * * * * * * * * * *
 57633        * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
```

Table 2c: Failures in the means test (. = 0, * = no test)
Chinese stations near (32^{o},110^{o})
10/01/90 to 10/31/90 at 00Z

STATION / Lvl	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
72456	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
72357	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
72349	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
72340	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
72260	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
72247	.	.	.	.	.	.	.	.	.	1	.	.	.	1	1	.
72239	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
72257	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*

Table 3a: Failures in the standard deviations test (. = 0, * = no test)
U.S. stations near (35^{o},-95^{o})
10/01/90 to 10/31/90 at 00Z

STATION / Lvl	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
26702	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
26406	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
12425	.	.	.	.	.	.	.	.	.	.	.	.	1	.	.	.
12374	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*
12330	.	1	.	.	.	.	.	.	.	.	.	.	.	.	*	*
12120	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*	*
10437	.	.	.	.	.	.	.	.	.	.	.	*	*	*	*	*
10338	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
10046	.	.	.	.	.	.	.	.	.	.	.	*	*	*	*	*
10035	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
9393	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
9184	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
6181	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
2591	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*
2527	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*	*

Table 3b: Failures in the standard deviations test (. = 0, * = no test)
European stations near (55^{o},15^{o})
10/01/90 to 10/31/90 at 00Z

STATION / Lvl	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
57679	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*
57516	*	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
57494	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
57461	.	*	.	.	.	.	.	.	.	.	.	.	.	.	.	*
57447	.	*	.	.	.	.	.	.	.	.	.	.	.	.	.	.
57328	*	*	.	.	.	.	.	*	*	*	*	*	*	*	*	*
57178	.	*	.	.	.	.	.	.	.	.	.	.	.	.	.	*
57127	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.
57083	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*
57036	.	.	.	.	.	.	.	.	.	.	.	.	.	.	.	*
57245	*	*	.	.	.	.	.	.	.	.	.	*	*	*	*	*
57633	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*

Table 3c: Failures in the standard deviations test (. = 0, * = no test)
Chinese stations near (32^{o},110^{o})
10/01/90 to 10/31/90 at 00Z

STATION / Lvl	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
72456	31	30	31	31	31	31	31	31	31	31	30	29	27	24	23	0
72357	31	30	31	31	31	30	31	31	31	30	30	30	29	26	22	0
72349	31	30	31	31	30	31	30	30	30	30	30	29	28	28	22	0
72340	31	29	31	31	30	30	30	30	30	30	29	28	28	26	22	0
72260	30	29	30	30	30	30	30	29	29	29	29	29	28	26	23	0
72247	31	28	31	31	31	31	31	31	31	31	29	29	29	27	23	0
72239	1	1	1	1	1	1	1	1	0	0	0	0	0	0	0	0
72257	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0

Table 4: Number of days daily outlier test was performed
U.S. stations near (35^{o},-95^{o})
10/01/90 to 10/31/90 at 00Z

STATION / Lvl	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
72456	31	30	31	31	31	31	31	31	31	31	30	29	28	25	24	11
72357	31	30	31	31	31	30	31	31	31	30	30	30	30	28	25	4
72349	31	30	31	31	30	31	30	30	30	30	30	29	28	28	23	6
72340	31	29	31	31	30	30	30	30	30	30	29	28	28	27	24	2
72260	30	29	30	30	30	30	30	29	29	29	29	29	29	28	27	3
72247	31	28	31	31	31	31	31	31	31	31	29	29	29	27	23	6
72239	1	1	1	1	1	1	1	1	0	0	0	0	0	0	0	0
72257	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0

Table 5: Number of daily reports
U.S. stations near (35^{o},-95^{o})
10/01/90 to 10/31/90 at 00Z

STATION 72456: LAT,LON = 39, -95

LEVEL	#DAYS	DAILY TESTS		MEANS TESTS		STD TESTS	TOTAL
1	31	PASS	PASS	PASS	PASS	PASS	PASS
2	30	PASS	PASS	PASS	PASS	PASS	PASS
3	31	PASS	PASS	PASS	PASS	PASS	PASS
4	31	PASS	*FAILH	PASS	PASS	PASS	*FAIL*
5	31	PASS	PASS	PASS	PASS	PASS	PASS
6	31	LFAIL*	PASS	PASS	PASS	PASS	*FAIL*
7	31	LFAIL*	PASS	PASS	PASS	PASS	*FAIL*
8	31	LFAIL*	*FAILH	PASS	PASS	PASS	*FAIL*
9	31	LFAIL*	PASS	PASS	PASS	PASS	*FAIL*
10	31	PASS	PASS	PASS	PASS	PASS	PASS
11	30	LFAIL*	PASS	PASS	PASS	PASS	*FAIL*
12	29	PASS	PASS	PASS	PASS	PASS	PASS
13	28	LFAIL*	PASS	PASS	PASS	PASS	*FAIL*
14	25	PASS	*FAILH	PASS	PASS	PASS	*FAIL*
15	24	LFAIL*	*FAILH	PASS	PASS	PASS	*FAIL*
16	11	NOTEST	NOTEST	PASS	PASS	PASS	PASS

Table 6: Summary report for station 72456
10/01/90 to 10/31/90 at 00Z

Appendix: Program listing

```
C       THIS PROGRAM IS FOR CHECKING THE STANDARD FORECAST DEVIATION
C       FILES FOR OUTLIERS IN:  (1)  DAILY READINGS, (2)  MEANS, AND
C       (3)  STANDARD DEVIATIONS OF STATIONS WITHIN A TARGET DISTANCE
C       OF A GIVEN LATITUDE AND LONGITUDE.
C
C       WRITTEN BY RICHARD FRANKE
C                  DEPARTMENT OF MATHEMATICS
C                  NAVAL POSTGRADUATE SCHOOL
C                  MONTEREY, CA  93943
C                  0083P@CC.NPS.NAVY.MIL
C                  408/646-2758
C
C       DATE OF BEGINNING:  6/21/91
C       DATES OF CHANGES:   6/21/91
C       MODIFIED TO USE SYMBOLS FOR OUTPUT:  6/21/91
C       MINOR PRETTYING UP:  6/24/91
C       MODIFIED TO NOTE NUMBER OF TESTS AND SUMMARY BY STATION: 6/27/911
C
        PARAMETER (MX=152,NS=15)
        REAL FE(MX,NS,16),FER(MX,NS),XMN(NS),STD(NS),FED(NS)
        INTEGER YRMD(MX,NS),NSTAT(NS),KNT(NS),NSD(NS),
     1  NDLF(NS,16),NDHF(NS,16),NSDF(NS,16),NMNLF(NS,16),NMNHF(NS,16),
     2  NSS(NS),YMD(MX,NS),NDG(NS,16),NDDT(NS,16),LATS(NS),LONS(NS)
        CHARACTER*15 IUNIT,OUNIT
        CHARACTER*6 FLAGS(-1:1),FLAGSL(-1:1),FLAGSH(-1:1)
        CHARACTER*1 SY(-1:36)
        CHARACTER*38 SYY
        EQUIVALENCE (SY,SYY)
        DATA SYY/'*.123456789ABCDEFGHIJKLMNOPQRSTUVWXYZ&'/
        DATA FLAGS/'NOTEST',' PASS ','*FAIL*'/,
     1     FLAGSL/'NOTEST',' PASS ','LFAIL*'/,
     2     FLAGSH/'NOTEST',' PASS ','*FAILH'/
C
C       GET DATA INPUT FILE NAME AND OUTPUT FILE NAME
C
        WRITE(*,9)
    9   FORMAT(' NAME THE INPUT FILE')
        READ(*,10)IUNIT
   10   FORMAT(A15)
        OPEN(4,BLANK='ZERO',FILE=IUNIT,MODE='READ',STATUS='OLD')
        WRITE(*,21)
   21   FORMAT(' NAME THE OUTPUT FILE')
        READ(*,10)OUNIT
        OPEN(8,FILE=OUNIT,MODE='WRITE',STATUS='NEW')
C
C       GET THE INPUT
C
  100   WRITE(*,1)
    1   FORMAT(' INPUT THE TARGET LATITUDE, LONGITUDE AND THE RANGE',/
     1 ' - FREE FIELD - RANGE = 0 STOPS EXECUTION',/)
        READ(*,*)LAT,LON,LLR
        IF(LLR.LE.0)STOP
```

```
C
C       STRETCH OUT IN THE LONGITUDINAL DIRECTION
C
        LTR=LLR/COS(FLOAT(LAT)/57.3)
C
C       START INPUT UNIT 4 AT THE BEGINNING
C
        REWIND 4
        MS = 0
C
C       READ A HEADER
C
  200   READ(4,2,END=300)NST,ILAT,ILON,ITM,NREC
    2   FORMAT(BZ,3X,I7,4X,2I5,8X,I2,I5)
        IF(ABS(ILAT-LAT).LE.LLR .AND. ABS(ILON-LON).LE.LTR) THEN
C
C         IT'S IN THE RECTANGLE, GET THE DATA
C
          MS=MS + 1
          NSTAT(MS)=NST
          LATS(MS)=ILAT
          LONS(MS)=ILON
          READ(4,3)((FE(I,MS,J),J=1,16),YRMD(I,MS),I=1,NREC)
    3     FORMAT(BZ,16F4.0,I8)
        ELSE
C
C         OR SKIP IT
C
          READ(4,4)(JUNK,I=1,NREC)
        ENDIF
    4   FORMAT(I1)
C
C       GO BACK FOR MORE IF WE DON'T HAVE THE MAX
C
        IF(MS.LT.NS) GO TO 200
        PRINT 7
    7   FORMAT(' STOPPING BEFORE END')
  300   CONTINUE
        IF(MS.LT.4) THEN
          WRITE(*,5)MS
C
C         NOT ENOUGH DATA, START OVER
C
    5     FORMAT(' ONLY',I2,' STATIONS FOUND - START OVER')
        GO TO 100
        ELSE
          WRITE(*,6)MS,(NSTAT(N),N=1,MS)
    6     FORMAT(I5,' STATIONS FOUND:',/(I10))
        ENDIF
C
C       GET THE DATE RANGE
C
  301   WRITE(*,8)
```

```
    8   FORMAT(' INPUT THE BEGINNING AND ENDING DAYS (>=1, <=152)'/
     1         ' BEGINNING = ZERO GOES TO NEW LATITUDE/LONGITUDE'/
     2         ' BEGINNING = NEGATIVE ENDS RUN')
        READ(*,*)IDB,IDE
        IF(IDB.LT.0)STOP
        IF(IDB.EQ.0)GO TO 100
C
C       OK, LET'S DO THE JOB
C
C       ZERO OUT FLAG FIELDS
C
        DO 304 L=1,16
          DO 303 N=1,NS
            NDLF(N,L) = 0
            NDHF(N,L) = 0
            NSDF(N,L) = -1
            NMNLF(N,L) = -1
            NMNHF(N,L) = -1
            NDG(N,L) = 0
            NDDT(N,L) = 0
  303       CONTINUE
  304     CONTINUE
C
C       LOOP OVER ALL LEVELS
C
        DO 400 L=1,16
          KS=0
C
C         KS COUNTS STATIONS WITH A HISTORY OF 2 OR MORE DAYS
C
          DO 305 N=1,MS
            KNT(N) = 0
            XMN(N) = 0.
C
C           KNT COUNTS HOW MANY DAYS
C
  305     CONTINUE
          DO 310 I=IDB,IDE
            ND=0
            DO 308 N=1,MS
C
C             ND COUNTS THE NUMBER OF STATIONS WITH GOOD DATA TODAY
C             FED SAVES IT FOR THE TEST
C             NSD TELLS WHICH STATIONS HAVE GOOD DATA
C             FER SAVES IT ALL FOR MEAN AND STD CALCS
C             YRMD SAVES THE DATE INFORMATION (NOT USED PRESENTLY, 6/21
C
              FED(N)=0.
              IF(FE(I,N,L).LT.999) THEN
                NDG(N,L)=NDG(N,L)+1
                KNT(N) = KNT(N) + 1
                ND=ND+1
                FED(ND)=FE(I,N,L)
                NSD(ND)=N
```

```
          FER(KNT(N),N) = FE(I,N,L)
          YMD(KNT(N),N) = YRMD(I,N)
          XMN(N)=XMN(N)+FER(KNT(N),N)
        ENDIF
  308   CONTINUE
C
C       TEST CURRENT DAY'S DATA
C
        CALL R11P05(FED,ND,NB,IB,NT,IT,IER)
        IF(IER.LE.0) THEN
          IF(NB.GT.0)NDLF(NSD(IB),L) = NDLF(NSD(IB),L) + 1
          IF(NT.GT.0)NDHF(NSD(IT),L) = NDHF(NSD(IT),L) + 1
          DO 309 N=1,ND
            NDDT(NSD(N),L) = NDDT(NSD(N),L) + 1
  309     CONTINUE
        ENDIF
  310 CONTINUE
      DO 315 N=1,MS
          IF(NDDT(N,L).EQ.0) THEN
            NDLF(N,L) = -1
            NDHF(N,L) = -1
          ENDIF
  315 CONTINUE
       DO 330 N=1,MS
         IF(KNT(N).GT.1)THEN
C
C        CALCULATE MEAN AND ST. DEV IF MORE THAN ONE DAY OF DATA
C
           KS=KS+1
           NSS(KS) = N
           XMN(KS)=XMN(N)/KNT(N)
          S= 0.
          DO 320 I=1,KNT(N)
            S=S+(FER(I,N)-XMN(KS))**2
  320     CONTINUE
        STD(KS) = SQRT(S/(KNT(N)-1))
        ENDIF
  330 CONTINUE
C
C     RUN TEST ON MEAN VALUES FOR THE STATIONS
C
      CALL R11P05(XMN,KS,NB,IB,NT,IT,IER)
      IF(IER.EQ.0)THEN
        DO 340 N=1,KS
          NMNLF(NSS(N),L) = 0
          NMNHF(NSS(N),L) = 0
  340        CONTINUE
        IF(NB.NE.0) NMNLF(NSS(IB),L)=1
        IF(NT.NE.0) NMNHF(NSS(IT),L)=1
      ENDIF
C
C     RUN TEST ON STANDARD DEVIATIONS OF THE STATIONS
C
      CALL R10P05(STD,KS,NT,IT,IER)
```

```
          IF(IER.EQ.0)THEN
            DO 360 N=1,KS
              NSDF(NSS(N),L) = 0
  360       CONTINUE
            IF(NT.GT.0)NSDF(NSS(IT),L) = 1
          ENDIF
  400   CONTINUE
        MDATB=0
        MDATE=0
        DO 410 N=1,MS
          IF(MDATB.EQ.0)MDATB=YRMD(IDB,N)
          IF(MDATE.EQ.0)MDATE=YRMD(IDE,N)
  410   CONTINUE
        IF(MDATB.EQ.0)MDATB=MDATE-(IDE-IDB)
        IF(MDATB.LT.0)MDATB=IDB
        IF(MDATE.EQ.0)MDATE=MDATB+(IDE-IDB)
C
C       OUTPUT RESULTS FOR VARIOUS TESTS
C
        WRITE(8,11)MDATB,MDATE,ITM,LAT,LON,LLR,LTR
   11   FORMAT('1  RESULTS OF THE OUTLIER TESTS FROM',
     1    I8,' TO',I8,' AT',I3,'Z'//
     2    '   NUMBER OF FAILURES BY LEVEL AND STATION: .=0, *=NOTEST'
     3    //'   TARGET LATITUDE/LONGITUDE IS',2I5,', RANGE:',I2,'X',I2
     4    //'   RESULTS OF THE DAILY OUTLIER TEST (LOW FAIL/HIGH FAIL)')
        WRITE(8,12)((L),L=1,16)
   12   FORMAT(/' STATION / Lvl',16I4/)
        WRITE(8,13)(NSTAT(N),
     1     (SY(NDLF(N,L)),SY(NDHF(N,L)),L=1,16),N=1,MS)
   13   FORMAT(I8,5X,32A2)
        WRITE(8,20)IDE-IDB+1
   20   FORMAT(//'   NUMBER OF DAYS DAILY OUTLIER TEST WAS PERFORMED'
     1    ' OUT OF',I4)
        WRITE(8,12)((L),L=1,16)
        WRITE(8,18)(NSTAT(N),(NDDT(N,L),L=1,16),N=1,MS)
   18   FORMAT(I8,4X,16I4)
        WRITE(8,14)
   14   FORMAT(//'   RESULTS OF THE MEANS OUTLIER TEST (LOW FAIL/HIGH'
     1         ' FAIL)')
        WRITE(8,12)((L),L=1,16)
        WRITE(8,13)(NSTAT(N),
     1     (SY(NMNLF(N,L)),SY(NMNHF(N,L)),L=1,16),N=1,MS)
        WRITE(8,15)
   15   FORMAT(//'   RESULTS OF THE STANDARD DEVIATION OUTLIER TEST')
        WRITE(8,12)((L),L=1,16)
        WRITE(8,16)(NSTAT(N),(SY(NSDF(N,L)),L=1,16),N=1,MS)
   16   FORMAT(I8,4X,16A4)
        WRITE(8,17)IDE-IDB+1
   17   FORMAT(//'   NUMBER OF DAYS REPORTING OUT OF',I4)
        WRITE(8,12)((L),L=1,16)
        WRITE(8,18)(NSTAT(N),(NDG(N,L),L=1,16),N=1,MS)
C
C       OUTPUT SUMMARY BY STATION
C
```

```
      WRITE(8,22)
 22   FORMAT(//' ************',
     1'**************************************************************'
     2 //' SUMMARY BY STATION FOLLOWS'/)
      DO 500 N=1,MS
        WRITE(8,23)NSTAT(N),LATS(N),LONS(N)
 23     FORMAT(//' --------',
     1'--------------------------------------------------------------'
     2    /' STATION',I6,':  LAT,LON = ',I4,',',I4/
     3   /' LEVEL #DAYS      DAILY TESTS         MEANS TESTS    ',
     4    ' STD TESTS     TOTAL'/)
      DO 420 L=1,16
C
C       SET CODES FOR TESTS:  -1 IS NOTEST, 0 IS PASS ALL, 1 IS FAIL
C                                                  ONE OR MORE
C       FOR THE TOTAL TEST, NOTEST MEANS NO TESTS PERFORMED,
C                         PASS MEANS PASS ALL TESTS PERFORMED (IF ANY)
C                         FAIL MEANS FAILED AT LEAST ONE
C
        NFL=MIN(NDLF(N,L),1)
        NFH=MIN(NDHF(N,L),1)
        NMFL=MIN(NMNLF(N,L),1)
        NMFH=MIN(NMNHF(N,L),1)
        NSF=MIN(NSDF(N,L),1)
        NTF=MAX(NFL,NFH,NMFL,NMFH,NSF)
        WRITE(8,24)L,NDG(N,L),FLAGSL(NFL),FLAGSH(NFH),
     1   FLAGSL(NMFL),FLAGSH(NMFH),FLAGS(NSF),FLAGS(NTF)
420     CONTINUE
 24     FORMAT(2I6,4X,A6,3X,A6,5X,A6,3X,A6,5X,A6,5X,A6)
500   CONTINUE
      WRITE(8,19)
 19   FORMAT(//' *****************',
     1'*********************************************************'/////)
      GO TO 301
      END

     SUBROUTINE R10P05(X,N,NOUTT,IT,IER)
     PARAMETER (MAX=30)
     DIMENSION X(N),SX(MAX),ISX(MAX)
     DIMENSION CV(30)
C
C    THIS SUBROUTINE DOES THE R10 TEST FOR OUTLIERS AT THE .05 CRITICAL
C    TEST LEVEL.
C
C    THE CV VALUES ARE THE CRITICAL VALUES FROM TABLE V.
C
     DATA CV/2*0,.941,.765,.642,.560,.507,.468,.437,.412,.392,.376,
    1 .361,.349,.338,.329,.320,.313,.306,.300,.295,.290,.285,.281,
    2 .277,.273,.269,.266,.263,.260/
C
C    INPUT VALUES ARE:  X - THE DATA TO BE TESTED
C                       N - NUMBER OF DATA POINTS
C
C    OUTPUT VALUES ARE:  NOUTT - FLAG FOR LARGEST VALUE
```

```
C                                 0 IF OK, 1 IF NOT
C                         IB    - LOCATION IN X ARRAY
C                         IER   - ERROR RETURN INDICATOR
C                                 0 IF ALL OK
C                                 1 IF N OUT OF RANGE 4 TO MAX, INCLUSIV
C                                     NO TEST PERFORMED
C
C
C     CHECK FOR SUITABLE N
C
      IER = 0
      NOUTT = 0
      IF(N.LT.3.OR.N.GT.MAX)THEN
        IER = 1
        RETURN
      ENDIF
      DO 100 I=1,N
        SX(I) = X(I)
        ISX(I) = I
  100 CONTINUE
C
C     SORT INTO INCREASING ORDER
C
      CALL SHSORT(SX,ISX,N)
      RT = 0.
      IF(SX(N)-SX(1) .NE. 0.) RT = (SX(N) - SX(N-1))/(SX(N) - SX(1))
      IF(RT.GT.CV(N))THEN
        NOUTT = 1
        IT = ISX(N)
      ENDIF
      RETURN
      END

      SUBROUTINE R11P05(X,N,NOUTB,IB,NOUTT,IT,IER)
      DIMENSION X(N),SX(30),ISX(30)
      DIMENSION CV(30)
C
C     THIS SUBROUTINE DOES THE R11 TEST FOR OUTLIERS AT THE .05 CRITICAL
C     TEST LEVEL.
C
C     THE CV VALUES ARE THE CRITICAL VALUES FROM TABLE V.
C
      DATA CV/3*0,.955,.807,.689,.610,.554,.512,.477,.450,.428,.410,
     1 .395,.381,.369,.359,.349,.341,.334,.327,.320,.314,.309,.304,
     1 .299,.295,.291,.287,.283/
C
C     INPUT VALUES ARE:  X - THE DATA TO BE TESTED
C                        N - NUMBER OF DATA POINTS
C
C     OUTPUT VALUES ARE:  NOUTB - FLAG FOR SMALLEST VALUE
C                                 0 IF OK, 1 IF NOT
C                         IB    - LOCATION IN X ARRAY
C                         NOUTT - FLAG FOR LARGEST VALUE
C                                 0 IF OK, 1 IF NOT
```

```
C                             IT     - LOCATION IN X ARRAY
C                             IER    - ERROR RETURN INDICATOR
C                                      0 IF ALL OK
C                                      1 IF N OUT OF RANGE 5 TO 30, INCLUSIVE
C
C
C     CHECK FOR SUITABLE N
C
      IER = 0
      IF(N.LT.4.OR.N.GT.30)THEN
        IER = 1
        RETURN
      ENDIF
      DO 100 I=1,N
        SX(I) = X(I)
        ISX(I) = I
  100 CONTINUE
C
C     SORT INTO INCREASING ORDER
C
      CALL SHSORT(SX,ISX,N)
      RB = 0.
      IF(SX(N-1)-SX(1) .NE. 0.) RB = (SX(2) - SX(1))/(SX(N-1) - SX(1))
      NOUTB = 0
      IF(RB.GT.CV(N))THEN
        NOUTB = 1
        IB = ISX(1)
      ENDIF
      RT = 0.
      IF(SX(N)-SX(2) .NE. 0.) RT = (SX(N) - SX(N-1))/(SX(N) - SX(2))
      NOUTT = 0
      IF(RT.GT.CV(N))THEN
        NOUTT = 1
        IT = ISX(N)
      ENDIF
      RETURN
      END

C    Included for completeness only - from NONIMSL library at NPS


C     ..............................................................
C
C  A. IDENTIFICATION:
C     TITLE:       NUMERICAL SORT
C     ID:          M1-NPG-SHSORT (F-IV)
C     PROGRAMMER:  R. BRUNELL
C     DATE:        MARCH 1968
C     MODIFIED:    DEC. 1973   BY L. NOLAN
C
C  B. PURPOSE:
C     TO SORT, IN ASCENDING ORDER, AN ARRAY OF SINGLE PRECISION REAL
C     NUMBERS BY THE METHOD OF SHELL, AND TO PRODUCE AN ARRAY OF INDEXES
C     SO USER CAN RE-ORDER OTHER CORRESPONDING INFORMATION ACCORDING TO
```

```
C     ASCENDING VALUES OF "A".
C
C  C. USAGE:
C     1. CALLING STATEMENT:
C        CALL SHSORT(A,KEY,N)
C     2. ARGUMENTS:
C          A - ARRAY OF NUMBERS TO BE SORTED. THIS ARRAY IS SORTED
C              (RE-ORDERED) BY "SHSORT".
C        KEY - ARRAY, DIMENSIONED AT LEAST N IN CALLING PROGRAM, TO BE
C              FILLED BY USER WITH INTEGERS FROM 1 TO N.  AFTER EXIT
C              FROM SHSORT, KEY(1) WILL CONTAIN THE ORIGINAL INDEX OF
C              THE SMALLEST ELEMENT OF "A"; KEY(2) WILL CONTAIN THE
C              ORIGINAL INDEX OF THE NEXT-TO-SMALLEST ELEMENT OF "A";
C              ETC.  KEY(N) WILL CONTAIN THE ORIGINAL INDEX OF THE
C              LARGEST ELEMENT OF "A".
C          N - NUMBER OF MEMBERS IN ARRAYS "A" AND "KEY".
C
C  D. REFERENCES:
C     1. "ALGORITHM 201, SHELLSORT", BOOTHROYD, J., "COMMUNICATIONS OF
C         ACM", VOL 6, NO 8, AUGUST 1963, P.445.
C     2. "CERTIFICATION OF ALGORITHM 201", BATTY,M.A., "COMMUNICATIONS
C         OF ACM", VOL 7, NO 6, JUNE 1964, P.349.
C
      SUBROUTINE SHSORT(A,KEY,N)
      DIMENSION A(N),KEY(N)
      M1=1
    6 M1=M1*2
      IF (M1 .LE. N)  GO TO 6
      M1=M1/2-1
      MM=MAX0(M1/2,1)
      GO TO 21
   20 MM=MM/2
      IF (MM .LE. 0)  GO TO 100
   21 K=N-MM
   22 DO 1 J=1,K
      II=J
   11 IM=II+MM
      IF (A(IM) .GE. A(II))  GO TO 1
      TEMP=A(II)
      IT=KEY(II)
      A(II)=A(IM)
      KEY(II)=KEY(IM)
      A(IM)=TEMP
      KEY(IM)=IT
      II=II-MM
      IF (II .GT. 0)  GO TO 11
    1 CONTINUE
      GO TO 20
  100 RETURN
      END
```

INITIAL DISTRIBUTION LIST

DIRECTOR OF RESEARCH (1)
ADMINISTRATION
CODE 08
NAVAL POSTGRADUATE SCHOOL
MONTEREY, CA 93943

DIRECTOR (2)
DEFENSE TECH. INFORMATION
CENTER, CAMERON STATION
ALEXANDRIA, VA 22314

LIBRARY (2)
NAVAL POSTGRADUATE SCHOOL
MONTEREY, CA 93943

DR. JOHN HOVERMALE
DIRECTOR, ATMOSPHERIC SCIENCES
DIVISION
NAVAL OCEANOGRAPHIC AND
ATMOSPHERIC RESEARCH LABORATORY
MONTEREY, CA 93943

COMMANDING OFFICER
NOARL
NSTL, MS 39529

COMMANDING OFFICER
FLENUMOCEANCEN
MONTEREY, CA 93943

NAVAL OCEANOGRAPHIC AND (5)
ATMOSPHERIC RESEARCH LABORATORY
ATTN: DR. EDWARD BARKER
DR. JAMES GOERSS
NANCY BAKER
PAT PHOEBUS
LARRY PHEGLEY
MONTEREY, CA 93943

PROFESSOR H. FREDRICKSEN
CHAIRMAN, DEPT OF MATHEMATICS
NAVAL POSTGRADUATE SCHOOL
MONTEREY, CA 93943

CDR. BOUNDS
CODE 22
OFFICE OF NAVAL TECHNOLOGY
800 N. QUINCY ST.
ARLINGTON, VA 22217

DR. H. JEAN THIEBAUX
NATIONAL METEOROLOGICAL CENTER
DEVELOPMENT DIVISION
WASHINGTON, DC 20233

DEPT. OF MATHEMATICS
NAVAL POSTGRADUATE SCHOOL
MONTEREY, CA 93943

NAVAL POSTGRADUATE SCHOOL
METEOROLOGY DEPT.
MONTEREY, CA 93943

DIRECTOR
OFFICER OF PROGRAMS RX3
NOAA RESEARCH LAB
BOULDER, CO 80302

DIRECTOR
NATIONAL SEVERE STORMS LAB
1313 HALLEY CIRCLE
NORMAN, OK 73069

PROFESSOR R. FRANKE, MA/FE (10)
DEPT. OF MATHEMATICS
NAVAL POSTGRADUATE SCHOOL
MONTEREY, CA 93943

PROFESSOR T. JAYACHANDRAN, MA/JY
DEPT. OF MATHEMATICS (10)
NAVAL POSTGRADUATE SCHOOL
MONTEREY, CA 93943

LIBRARY, FLEET NUMERICAL (2)
OCEANOGRAPHY CENTER
MONTEREY, CA 93943